

ASP502

:

“

”

```
/**  
 * @name  
 * @email cybaek@netsgo.com  
 * @homepage http://cybaek.com/  
 */
```



ASP502	:	“	”	1
				4
				4
				4
				4
				5
				6
				6
				: /admin/ListLinkArticles.asp	6
				6
				6
				: /admin/ListLinkArticles.asp	10
CDebug				11
				11
				: /admin/LinkArticleInsertForm.asp	12
				12
				12
				: /admin/LinkArticleInsertForm.asp.....	12
				: <form>	13
				13
				: <form>	13
				: /admin/LinkArticleInsertAction.asp.....	14
				14
				14
				: /admin/LinkArticleInsertAction.asp	14
				: /admin/LinkArticleDeleteAction.asp.....	15
				15
				15
				: /admin/LinkArticleDeleteAction.asp.....	15
				: /admin/LinkArticleUpdateForm.asp	15
				15
				16
				: /admin/LinkArticleUpdateForm.asp	16
				: /admin/LinkArticleUpdateAction.asp	17
				17
				18
				: /admin/LinkArticleUpdateAction.asp.....	18
				18
				21
				21

: /admin/ArticleList.asp.....	21
.....	21
: /admin/ArticleList.asp.....	21
: GoodNews.Article	23
: /admin/C.ArticleList.asp	24
.....	24
.....	24
: /admin/C.ArticleList.asp	25
: /admin/ArticleContent.asp	26
.....	26
.....	26
: /admin/ArticleContent.asp	26
: /admin/C.ArticleContent.asp	27
.....	27
.....	27
: /admin/C.ArticleContent.asp.....	28
: /admin/C.ArticleUpdateAction.asp.....	29
.....	29
.....	29
: /admin/C.ArticleUpdateAction.asp	30

가

•

•

•

•

.

•

•

•

.

.

•

•

```

CREATE TABLE [dbo].[Articles] (
    [ArticleId] [int] IDENTITY (1, 1) NOT NULL ,
    [Title] [varchar] (150) COLLATE Korean_Wansung_CI_AS NULL ,
    [SubTitle] [varchar] (200) COLLATE Korean_Wansung_CI_AS NULL ,
    [Reporter] [varchar] (20) COLLATE Korean_Wansung_CI_AS NULL ,
    [InputDate] [datetime] NULL ,
    [Content] [text] COLLATE Korean_Wansung_CI_AS NULL
) ON [PRIMARY] TEXTIMAGE_ON [PRIMARY]
GO

CREATE TABLE [dbo].[Headlines] (
    [HeadlineId] [int] IDENTITY (1, 1) NOT NULL ,
    [ArticleId] [int] NULL ,
    [Summary] [varchar] (400) COLLATE Korean_Wansung_CI_AS NULL ,
    [InputDate] [datetime] NULL
) ON [PRIMARY]
GO

CREATE TABLE [dbo].[LinkArticles] (
    [LinkArticleId] [int] IDENTITY (1, 1) NOT NULL ,
    [Url] [varchar] (200) COLLATE Korean_Wansung_CI_AS NULL ,
    [Title] [varchar] (150) COLLATE Korean_Wansung_CI_AS NULL ,
    [InputDate] [datetime] NULL
) ON [PRIMARY]
GO

CREATE TABLE [dbo].[Memos] (
    [MemoId] [int] IDENTITY (1, 1) NOT NULL ,
    [ArticleId] [int] NOT NULL ,
    [Memo] [varchar] (150) COLLATE Korean_Wansung_CI_AS NULL ,
    [Writer] [varchar] (20) COLLATE Korean_Wansung_CI_AS NULL ,
    [InputDate] [datetime] NULL
) ON [PRIMARY]
GO

```

GoodNews.com



DBHelper, <form>

,

.

:/admin/ListLinkArticles.asp



LinkArticles

.



가

.



: 30 .



.



: 15

LinkArticleList.asp

```
<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

' @      strSQL      SQL
' strSQL = "SELECT ..."

' @      DBHelper      rs
' Set rs =

all = tpl.Load(SITE_HOME & "\admin\tpl.LinkArticleList.html")

strArea = tpl.GetArea(all, "header")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

Do While (Not rs.EOF) And (Not rs.BOF)
    strArea = tpl.GetArea(all, "item")
    dic.RemoveAll

    dic.Add "LinkArticleId", CStr2(rs("LinkArticleId"))
```

```

        ' @
        dic
        가
        .

        strCvt = tpl.Convert(strArea, dic)
        Response.Write strCvt

        rs.MoveNext
Loop

strArea = tpl.GetArea(all, "footer")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

rs.Close
Set rs = Nothing

Call FreeObjects()
%>

```

Connection.asp

```

<%
Const CONN = "File Name=d:\Config\www.goodnews.com.udl"
%>

```

Const.asp

```

<%
Const SITE_HOME = "d:\WebSites\www.goodnews.com"
%>

```

CDebug.asp

```

<%
Class CDebug
    Dim mb_Enabled
    Dim md_RequestTime
    Dim md_FinishTime
    Dim mo_Storage

    Public Default Property Get Enabled()
        Enabled = mb_Enabled
    End Property

    Public Property Let Enabled(bNewValue)
        mb_Enabled = bNewValue
    End Property

    Private Sub Class_Initialize()
        md_RequestTime = Now()
        Set mo_Storage = Server.CreateObject("Scripting.Dictionary")
    End Sub

    Public Sub Print(label, output)
        If Enabled Then

```

```

        Call mo_Storage.Add(label, output)
    End If
End Sub

Public Sub [End]()
    md_FinishTime = Now()
    If Enabled Then
        Response.Write "<p><span class='trace__'>" & vbCrLf
        Call PrintSummaryInfo()
        Call PrintCollection("VARIABLE STORAGE", mo_Storage)
        Call PrintCollection("QUERYSTRING COLLECTION", Request.QueryString())
        Call PrintCollection("FORM COLLECTION", Request.Form())
        Call PrintCollection("COOKIES COLLECTION", Request.Cookies())
        Call PrintCollection("SERVER VARIABLES COLLECTION", & _
            Request.ServerVariables())
        Response.Write "</span>"
    End If
End Sub

Private Sub PrintSummaryInfo()
    Dim i

    PrintTableHeader("SUMMARY INFO")

    Response.Write("<tr><td>Start Time of Request</td><td>" & _
        md_RequestTime & "</td></tr>" & vbCrLf)
    Response.Write("<tr class='alt'><td>Finish Time of Request</td><td>" & _
        md_FinishTime & "</td></tr>" & vbCrLf)
    Response.Write("<tr><td>Elapsed Time</td><td>" & _
        DateDiff("s", md_RequestTime, md_FinishTime) & "</td></tr>" & _
        vbCrLf)
    Response.Write("<tr class='alt'><td>Request Type</td><td>" & _
        Request.ServerVariables("REQUEST_METHOD") & "</td></tr>" & vbCrLf)
    Response.Write("<tr><td>Status Code</td><td>" & Response.Status & _
        "</td></tr>" & vbCrLf)

    Response.Write "</tr></table>"
End Sub

Private Sub PrintCollection(Byval Name, ByVal Collection)
    Dim vItem
    Dim i

    PrintTableHeader(Name)

    For Each vItem In Collection
        If i mod 2 = 0 Then
            Response.Write("<tr>")
        else
            Response.Write("<tr class='alt'>")
        end if
        Response.Write("<td>" & vItem & "</td><td>" & _
            Collection(vItem) & "</td></tr>" & vbCrLf)
        i = i + 1
    Next

    Response.Write "</tr></table>"
End Sub

```



```

Private Sub Class_Terminate()
    Set mo_Storage = Nothing
End Sub

Private Sub PrintTableHeader(ByVal Name)
    Response.Write "<table cellpadding='0' width='100%' cellspacing='0'>" & _
        vbCrLf
    Response.Write "<tr><th class='alt' colspan='10' align='left'>" & _
        "<h3><b>" & Name & "</b></h3></th></tr>" & vbCrLf
    Response.Write "<tr class='subhead' align='left'>" & _
        "<th width='10%'>Name</th><th width='10%'>Value</th></tr>" & _
        vbCrLf
End Sub
End Class
%>

```

Declare.asp

```

<!--METADATA TYPE="typelib"
    FILE="C:\Program Files\Common Files\System\ado\msado15.dll"-->
<!--METADATA TYPE="typelib"
    FILE="c:\winnt\system32\scrrun.dll"-->
<OBJECT PROGID=Scripting.FileSystemObject id=fso RUNAT=server></OBJECT>
<OBJECT PROGID=DAL.DBHelper id=dbh RUNAT=server></OBJECT>
<OBJECT PROGID=UI.TPL id=tpl RUNAT=server></OBJECT>
<OBJECT PROGID=Scripting.Dictionary id=dic RUNAT=server></OBJECT>
<%
Dim all, strArea, strCvt
Dim strSQL, rs, rs2
Dim Debug

Set Debug = New CDebug
%>

```

Utils.asp

```

<%
Function CStr2(str)
    CStr2 = Replace(CStr(str & ""), Chr(0), "")
End Function

Function qq(str)
    qq = Replace(CStr2(str), "'", "'")
End Function

Function CInputValue(str)
    CInputValue = Replace(Replace(CStr2(str), "&", "&"), "'", "&quot;")
End Function

Function CTextareaValue(str)
    CTextareaValue = Replace(CStr2(str), "</textarea", "&lt;/textarea")
End Function

Function FreeObjects()
    Debug.End

```

```

        Set Debug = Nothing
End Function

Function RemovePTag(str)
    If (Left(str, 3) = "<P>" Or Left(str, 3) = "<p>") And _
        (Right(str, 4) = "</P>" Or Right(str, 4) = "</p>")Then
        str = Mid(str, 4, Len(str)-7)
    End If
    RemovePTag = str
End Function
%>

```



: /admin/ListLinkArticles.asp

```

ListLinkArticles.asp

<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

strSQL = "SELECT LinkArticleId, Title, Url, InputDate " & _
        " FROM LinkArticles (nolog) order by InputDate desc"
Set rs = dbh.RunSQLReturnRS(strSQL, "", CONN)

all = tpl.Load(SITE_HOME & "\admin\tpl.LinkArticleList.html")

strArea = tpl.GetArea(all, "header")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

Do While (Not rs.EOF) And (Not rs.BOF)
    strArea = tpl.GetArea(all, "item")
    dic.RemoveAll

    dic.Add "LinkArticleId", CStr2(rs("LinkArticleId"))
    dic.Add "Title", CStr2(rs("Title"))
    dic.Add "Url", CStr2(rs("Url"))
    dic.Add "InputDate", CStr2(rs("InputDate"))

    strCvt = tpl.Convert(strArea, dic)
    Response.Write strCvt

    rs.MoveNext

```

```

Loop

strArea = tpl.GetArea(all, "footer")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

rs.Close
Set rs = Nothing

Call FreeObjects()
%>

```

CDebug

True

```
Debug.Enabled = False
```



LinkArticleInsertForm.asp LinkArticleInsertAction.asp
 tpl.LinkArticleInsertForm.html

ASP

“ Form.asp”, ASP Action
 “ Action.asp”

:/admin/LinkArticleInsertForm.asp



-
- ☐ ASP
 - ☐ ListLinkArticles.asp
 - ☐ : 15

:/admin/LinkArticleInsertForm.asp

LinkArticleInsertForm.asp

```
<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

all = tpl.Load(SITE_HOME & "\admin\tpl.LinkArticleInsertForm.html")

strArea = tpl.GetArea(all, "header")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

Call FreeObjects()
%>
```

tpl.LinkArticleInsertForm.html

```
<!--{area,begin,header}-->
<link rel=stylesheet href=/site.css type=text/css>
<link rel=stylesheet href=/debug.css type=text/css>

<div class=header1>                </div>
<br>
```

```


<table width=100%>
<form action="LinkArticleInsertAction.asp" method=post id=form1 name=form1>
<tr>
    <th>    </th>
    <td><input type=text name=Title size=50></td>
</tr>
<tr>
    <th>    </th>
    <td><input type=text name=Url size=50></td>
</tr>
<tr>
    <th>&nbsp;</th>
    <td><input type=submit value="        "></td>
</tr>
</form>
</table>
<!--{area,end,header}-->

```

: <form>

❏ cybaek_validation.js

tpl.LinkArticleInsertForm.html



: <form>

tpl.LinkArticleInsertForm.html

```

<!--{area,begin,header}-->
<link rel=stylesheet href=/site.css type=text/css>
<link rel=stylesheet href=/debug.css type=text/css>
<script src="/cybaek_validation.js"></script>
<div class=header1>    </div>
<br>
<table width=100%>
<form action="LinkArticleInsertAction.asp" method=post
    onSubmit="return checkValue(this);">
<tr>
    <th>    </th>
    <td><input type=text name=Title size=50
        V=true
        MinLen=10
        MaxLen=150></td>

```

```

</tr>
<tr>
    <th>    </th>
    <td><input type=text name=Url size=50
        V=true
        MinLen=1
        MaxLen=200></td>
</tr>
<tr>
    <th>&nbsp;</th>
    <td><input type=submit value="    "></td>
</tr>
</form>
</table>
<!--{area,end,header}-->

```

: /admin/LinkArticleInsertAction.asp

☐
☐

INSERT

LinkArticlesList.asp

☐

:15

☐

: /admin/LinkArticleInsertAction.asp

LinkArticleInsertAction.asp

```

<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

Dim strTitle, strUrl
strTitle = Request("Title")
strUrl = Request("Url")

strSQL = "INSERT INTO LinkArticles (Title, Url) VALUES (?, ?)"

Call dbh.RunSQL(strSQL, _
    Array( _
        dbh.mp("@title", adVarChar, 150, strTitle), _
        dbh.mp("@url", adVarChar, 200, strUrl)), _

```

```
CONN)
```

```
Call FreeObjects()  
Response.Redirect "LinkArticleList.asp"  
%>
```

: /admin/LinkArticleDeleteAction.asp

<input type="checkbox"/>	LinkArticleId	.
<input type="checkbox"/>	, LinkArticlesList.asp	.
<input type="checkbox"/>	: 5	.

<input type="checkbox"/>	LinkArticleInsertAction.asp	SQL	.
--------------------------	-----------------------------	-----	---

: /admin/LinkArticleDeleteAction.asp

LinkArticleDeleteAction.asp


```
<%  
Option Explicit  
%>  
<!--#include virtual="/Common/Connection.asp" -->  
<!--#include virtual="/Common/Const.asp" -->  
<!--#include virtual="/Common/CDebug.asp" -->  
<!--#include virtual="/Common/Declare.asp" -->  
<!--#include virtual="/Common/Utils.asp" -->  
<%  
Debug.Enabled = False  
  
Dim strLinkArticleId  
strLinkArticleId = Request("LinkArticleId")  
  
strSQL = "DELETE LinkArticles WHERE LinkArticleId = ?"  
  
Call dbh.RunSQL(strSQL, _  
    Array(dbh.mp("@id", adInteger, 4, strLinkArticleId)), CONN)  
  
Call FreeObjects()  
Response.Redirect "LinkArticleList.asp"  
%>
```

: /admin/LinkArticleUpdateForm.asp

<input type="checkbox"/>	.
--------------------------	---



: 15 .

☐ LinkArticlesList.asp .☐ 1: . , SQL WHERE 가☐ 2: . , .☐ <input>☐ <input type=hidden>

:/admin/LinkArticleUpdateForm.asp

LinkArticleUpdateForm.asp

```
<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

Dim strLinkArticleId
strLinkArticleId = Request("LinkArticleId")

strSQL = "SELECT LinkArticleId, Title, Url " & _
        " FROM LinkArticles (nolock) WHERE LinkArticleId = ? "
Set rs = dbh.RunSQLReturnRS(strSQL, _
        Array(dbh.mp("@LinkArticleId", adInteger, 4, strLinkArticleId)), _
        CONN)

all = tpl.Load(SITE_HOME & "\admin\tpl.LinkArticleUpdateForm.html")

strArea = tpl.GetArea(all, "header")

dic.Add "LinkArticleId", CStr2(rs("LinkArticleId"))
```



```
dic.Add "Title", CInputValue(rs("Title"))
dic.Add "Url", CInputValue(rs("Url"))
```

```
rs.Close
Set rs = Nothing
```

```
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt
```

```
Call FreeObjects()
```

```
%>
```

```
tpl.HeadlineUpdateForm.html
```

```
<!--{area,begin,header}-->
<link rel=stylesheet href=/site.css type=text/css>
<link rel=stylesheet href=/debug.css type=text/css>
<script src="/cybaek_validation.js"></script>

<div class=header1>          </div>
<br>
<table width=100%>
<form action="LinkArticleUpdateAction.asp" method=post
      onsubmit="return checkValue(this);">
<input type=hidden name=LinkArticleId value="{LinkArticleId}">
<tr>
      <th>          </th>
      <td><input type=text name=Title size=50 value="{Title}"
              V=true
              MinLen=10
              MaxLen=150></td>
</tr>
<tr>
      <th>          </th>
      <td><input type=text name=Url size=50 value="{Url}"
              V=true
              MinLen=1
              MaxLen=200></td>
</tr>
<tr>
      <th>&nbsp;   </th>
      <td><input type=submit value="          "></td>
</tr>
</form>
</table>
<!--{area,end,header}-->
```

: /admin/LinkArticleUpdateAction.asp



, LinkArticlesList.asp

□ :5

□ SQL , LinkArticleDeleteAction.asp .

: /admin/LinkArticleUpdateAction.asp

LinkArticleUpdateAction.asp

```
<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

strSQL = "UPDATE LinkArticles SET Title = ?, Url = ? WHERE LinkArticleId = ?"

Dim strTitle, strUrl, strLinkArticleId
strTitle = Request("Title")
strUrl = Request("Url")
strLinkArticleId = Request("LinkArticleId")

Call dbh.RunSQL(strSQL, _
    Array( _
        dbh.mp("@title", adVarChar, 150, strTitle), _
        dbh.mp("@url", adVarChar, 200, strUrl), _
        dbh.mp("@LinkArticleId", adInteger, 4, strLinkArticleId)), _
    CONN)

Call FreeObjects()
Response.Redirect "LinkArticleList.asp"
%>
```

HTML

LinkArticlesList.asp

GenerateLinkArticleListHtml.asp

```
<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
```

```

<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

strSQL = "SELECT LinkArticleId, Title, Url, InputDate " & _
        " FROM LinkArticles (nolog)"
Set rs = dbh.RunSQLReturnRS(strSQL, "", CONN)

all = tpl.Load(SITE_HOME & "\admin\tpl.GenerateLinkArticleListHtml.html")

Dim strHtml
strArea = tpl.GetArea(all, "header")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt
strHtml = strCvt

Dim tmp
Do While (Not rs.EOF) And (Not rs.BOF)
    strArea = tpl.GetArea(all, "item")
    dic.RemoveAll

    dic.Add "LinkArticleId", CStr2(rs("LinkArticleId"))
    dic.Add "Title", CStr2(rs("Title"))
    dic.Add "Url", CStr2(rs("Url"))

    tmp = CStr2(rs("InputDate"))
    dic.Add "InputDate", FormatDateTime(tmp, 2) & " " & FormatDateTime(tmp, 4)

    strCvt = tpl.Convert(strArea, dic)
    Response.Write strCvt
    strHtml = strHtml & strCvt

    rs.MoveNext
Loop

strArea = tpl.GetArea(all, "footer")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt
strHtml = strHtml & strCvt

rs.Close
Set rs = Nothing

Dim strArticleListHtml, ts
strArticleListHtml = SITE_HOME & "\LinkArticleListHtml.html"
Set ts = fso.CreateTextFile(strArticleListHtml, True)
ts.Write(strHtml)
ts.Close
Set ts = Nothing

strArea = tpl.GetArea(all, "ok")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

Call FreeObjects()

```

%>

tpl.GenerateLinkArticleListHtml.html

```
<!--{area,begin,header}-->
<div class=ArticleListHeader>                                </div>
<table width=100%>
<!--{area,end,header}-->

<!--{area,begin,item}-->
<tr>
    <td class="ArticleListTitleItem"><li><a class="ArticleListLinkItem"
href="{Url}" target="NewWin${LinkArticleId}">${Title}</a>&nbsp;</td>
    <td class="ArticleListInputDateItem">${InputDate}&nbsp;</td>
</tr>
<!--{area,end,item}-->

<!--{area,begin,footer}-->
</table>

<!--{area,end,footer}-->

<!--{area,begin,ok}-->
<script>
alert('                .');
</script>
<!--{area,end,ok}-->
```

LinkArticleListHtml.html

```
<div class=ArticleListHeader>                                </div>
<table width=100%><tr>
    <td class="ArticleListTitleItem"><li><a class="ArticleListLinkItem"
href="http://news.nate.com/Service/natenews/ShellView.asp?LinkID=1&ArticleID=2003112
709311700397" target="NewWin1">                                100 弗
    (    )</a>&nbsp;</td>
    <td class="ArticleListInputDateItem">2003-11-27 10:58&nbsp;</td>
</tr></table>
```



ASP

,

WSC

.

:/admin/ArticleList.asp

Articles

.



:30 .

:/admin/ArticleList.asp

ArticleList.asp

```
<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

strSQL = "SELECT ArticleId, Title, SubTitle, Reporter, InputDate, Content " & _
        " FROM Articles (nolock) order by InputDate desc"
Set rs = dbh.RunSQLReturnRS(strSQL, "", strConnection)

all = tpl.Load(SITE_HOME & "\admin\tpl.ArticleList.html")

strArea = tpl.GetArea(all, "header")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

Do While (Not rs.EOF) And (Not rs.BOF)
    strArea = tpl.GetArea(all, "item")
    dic.RemoveAll

    dic.Add "ArticleId", CStr2(rs("ArticleId"))
    dic.Add "Title", CStr2(rs("Title"))
    dic.Add "SubTitle", CStr2(rs("SubTitle"))
    dic.Add "Reporter", CStr2(rs("Reporter"))
    dic.Add "InputDate", CStr2(rs("InputDate"))
    dic.Add "Content", CStr2(rs("Content"))

    strCvt = tpl.Convert(strArea, dic)
```

```

        Response.Write strCvt

        rs.MoveNext
Loop

strArea = tpl.GetArea(all, "footer")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

rs.Close
Set rs = Nothing

Call FreeObjects()
%>

```

```

tpl.ArticleList.html

<!--{area,begin,header}-->
<link rel=stylesheet href=/site.css type=text/css>
<link rel=stylesheet href=/debug.css type=text/css>
<script src="/goodnews.js" language="jscript"></script>

<div class=header1>          </div>
<br>
<table width=100%>
<tr>
    <th>#</th>
    <th>    </th>
    <th>    </th>
    <th>    </th>
    <th>    </th>
    <th>&nbsp;</th>
</tr>
<!--{area,end,header}-->

<!--{area,begin,item}-->
<tr>
    <td>${ArticleId}&nbsp;</td>
    <td><a href="" onclick="readArticle(${ArticleId}); return
false;">${Title}</a>&nbsp;</td>
    <td>${SubTitle}&nbsp;</td>
    <td>${Reporter}&nbsp;</td>
    <td>${InputDate}&nbsp;</td>
    <td>[<a href="ArticleUpdateForm.asp?ArticleId=${ArticleId}">    </a>]
    [<a
href="C.ArticleUpdateForm.asp?ArticleId=${ArticleId}">    (C)</a>]<br>
    [<a href="ArticleDeleteAction.asp?ArticleId=${ArticleId}"
    onclick="return(confirm('                ?'));">    </a>]
    [<a href="C.ArticleDeleteAction.asp?ArticleId=${ArticleId}"
    onclick="return(confirm('                ?'));">    (C)</a>]
    </td>
</tr>
<!--{area,end,item}-->

<!--{area,begin,footer}-->
</table>

```

```
<!--{area,end,footer}-->
```



: GoodNews.Article

WSC ProgId GoodNews.Article,
GoodNews.Article.WSC가 .

GetArticles 가 .

GoodNews.Article

```
<?xml version="1.0" encoding="euc-kr"?>
<component>

<?component error="true" debug="true"?>

<registration
  description="GoodNews Article"
  progid="GoodNews.Article"
  version="1.1"
  classid="{f86d6456-2ff6-45f6-b8a2-57abe53e7777}"
>
<script language="VBScript">
<![CDATA[
Function register()
  Dim tl
  Set tl = CreateObject("Scriptlet.TypeLib")
  tl.AddURL "GoodNews.Article.wsc"
  tl.Path = "GoodNews.Article.tlb"
  tl.GUID = "{f86d6456-2ff6-45f6-b8a2-57abe53e7778}"
  tl.Doc = "GoodNews.Article Class"
  tl.Name = "GoodNews.Article"
  tl.MajorVersion = 1
  tl.MinorVersion = 1
  tl.Write
  tl.Reset
  Set tl = Nothing

  Dim fso
  Set fso = CreateObject("Scripting.FileSystemObject")
  fso.CopyFile "GoodNews.Article.wsc", "GoodNews.Article.wsc.xml"
  Set fso = Nothing
End Function
]]>
</script>
</registration>

<reference object="Scripting.FileSystemObject"/>
<reference object="DAL.DBHelper"/>
```

```

<public>
    <method name="GetArticles">
        </method>
    </public>

<implements type="ASP" id="ASP"/>

<script language="VBScript">
<![CDATA[

Const DB_CON = "File Name=d:\Config\www.goodnews.com.UDL"
Const adInteger = 3

Function GetArticles()
    Dim dbh, strSQL

    strSQL = "SELECT ArticleId, Title, SubTitle, Reporter, " & _
        " InputDate, Content " & _
        " FROM Articles (nolog) order by InputDate desc"

    Set dbh = Server.CreateObject("DAL.DBHelper")
    Set GetArticles = dbh.RunSQLReturnRS(strSQL, "", DB_CON)

    Set dbh = Nothing
End Function

' private functions -----

Function CStr2(str)
    CStr2 = Replace(CStr(str & ""), Chr(0), "")
End Function

Function qq(str)
    qq = Replace(CStr2(str), "'", "'")
End Function
]]>
</script>

</component>

```

WSC

: /admin/C.ArticleList.asp



:5



ArticleList.asp

: /admin/C.ArticleList.asp

C.ArticleList.asp

```
<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

Set rs = article.GetArticles()

all = tpl.Load(SITE_HOME & "\admin\tpl.ArticleList.html")

strArea = tpl.GetArea(all, "header")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

Do While (Not rs.EOF) And (Not rs.BOF)
    strArea = tpl.GetArea(all, "item")
    dic.RemoveAll

    dic.Add "ArticleId", CStr2(rs("ArticleId"))
    dic.Add "Title", CStr2(rs("Title"))
    dic.Add "SubTitle", CStr2(rs("SubTitle"))
    dic.Add "Reporter", CStr2(rs("Reporter"))
    dic.Add "InputDate", CStr2(rs("InputDate"))
    dic.Add "Content", CStr2(rs("Content"))

    strCvt = tpl.Convert(strArea, dic)
    Response.Write strCvt

    rs.MoveNext
Loop

strArea = tpl.GetArea(all, "footer")
strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

rs.Close
Set rs = Nothing

Call FreeObjects()
%>
```

Declare.asp

```
<!--METADATA TYPE="typelib"
    FILE="C:\Program Files\Common Files\System\ado\msado15.dll"-->
<!--METADATA TYPE="typelib"
    FILE="c:\winnt\system32\scrrun.dll"-->
```

```

<OBJECT PROGID=Scripting.FileSystemObject id=fso RUNAT=server></OBJECT>
<OBJECT PROGID=DAL.DBHelper id=dbh RUNAT=server></OBJECT>
<OBJECT PROGID=UI.TPL id=tpl RUNAT=server></OBJECT>
<OBJECT PROGID=Scripting.Dictionary id=dic RUNAT=server></OBJECT>
<OBJECT PROGID=GoodNews.Article id=article RUNAT=server></OBJECT>
<%
Dim all, strArea, strCvt
Dim strSQL, rs, rs2
Dim Debug

Set Debug = New CDebug
%>

```

:/admin/ArticleContent.asp

<input type="checkbox"/>	.
<input type="checkbox"/> ArticleId	ID .
<input type="checkbox"/>	: 15 .

<input type="checkbox"/> ArticleList.asp	가 ,
--	-----

:/admin/ArticleContent.asp

ArticleContent.asp

```

<%
Option Explicit
%>
<!--#include virtual="/Common/Connection.asp" -->
<!--#include virtual="/Common/Const.asp" -->
<!--#include virtual="/Common/CDebug.asp" -->
<!--#include virtual="/Common/Declare.asp" -->
<!--#include virtual="/Common/Utils.asp" -->
<%
Debug.Enabled = False

Dim strArticleId
strArticleId = Request("ArticleId")

strSQL = "SELECT ArticleId, Title, SubTitle, Reporter, InputDate, Content " & _
        " FROM Articles (nolock) WHERE ArticleId = ? " & _
        " ORDER BY InputDate DESC"
Set rs = dbh.RunSQLReturnRS(strSQL, Array(dbh.mp("@id", adInteger, 4, strArticleId)), CONN)

all = tpl.Load(SITE_HOME & "\tpl.ArticleContent.html")

strArea = tpl.GetArea(all, "header")

```

```

dic.Add "ArticleId", CStr2(rs("ArticleId"))
dic.Add "Title", CStr2(rs("Title"))
dic.Add "SubTitle", CStr2(rs("SubTitle"))
dic.Add "Reporter", CStr2(rs("Reporter"))
dic.Add "Content", CStr2(rs("Content"))
dic.Add "InputDate", CStr2(rs("InputDate"))

rs.Close
Set rs = Nothing

strCvt = tpl.Convert(strArea, dic)
Response.Write strCvt

Call FreeObjects()
%>

```

:/admin/C.ArticleContent.asp

-
- ☐ WSC GetArticleByArticleId 가 .
 - ☐ C.ArticleContent.asp WSC .
 - ☐ : 15 .
-

- ☐ WSC .

```

GoodNews.Article.WSC

...

<reference object="Scripting.FileSystemObject"/>
<reference object="DAL.DBHelper"/>

<public>
    <method name="GetArticles">
    </method>
    <method name="GetArticleByArticleId">
        <PARAMETER name="ArticleId"/>
    </method>
</public>

<implements type="ASP" id="ASP"/>

<script language="VBScript">
<![CDATA[

Const DB_CON = "File Name=d:\Config\www.goodnews.com.UDL"
Const adInteger = 3

Function GetArticles()
    Dim dbh, strSQL

```

```

        strSQL = "SELECT ArticleId, Title, SubTitle, Reporter, InputDate, Content " &
        " FROM Articles (nolog) order by InputDate desc"

        Set dbh = Server.CreateObject("DAL.DBHelper")
        Set GetArticles = dbh.RunSQLReturnRS(strSQL, "", DB_CON)

        Set dbh = Nothing
End Function

Function GetArticleByArticleId(strArticleId)

End Function

' private functions -----
...

```

: /admin/C.ArticleContent.asp

```

C.ArticleContent.asp
...

<%
Debug.Enabled = False

Dim strArticleId
strArticleId = Request("ArticleId")

Set rs = article.GetArticleByArticleId(strArticleId)

all = tpl.Load(SITE_HOME & "\tpl.ArticleContent.html")

strArea = tpl.GetArea(all, "header")

...

```

```

GoodNews.Article.wsc
...

Function GetArticleByArticleId(strArticleId)
    Dim dbh, strSQL

    strSQL = "SELECT ArticleId, Title, SubTitle, Reporter, " & _
        " InputDate, Content " & _
        " FROM Articles (nolog) WHERE ArticleId = ? " & _
        " ORDER BY InputDate DESC"

    Set dbh = Server.CreateObject("DAL.DBHelper")
    Set GetArticleByArticleId = dbh.RunSQLReturnRS( _
        strSQL, _
        Array(dbh.mp("@ArticleId", adInteger, 4, strArticleId)), DB_CON)

```

```

        Set dbh = Nothing
End Function

...

```

: /admin/C.ArticleUpdateAction.asp

☐ WSC UpdateArticleById 가 .
☐ C.ArticleUpdateAction.asp WSC .
☐ :25 .

☐ .
☐ .

GoodNews.Artice.wsc

```

...

<public>
    <method name="GetArticles">
    </method>
    <method name="GetArticleById">
        <PARAMETER name="ArticleId"/>
    </method>
    <method name="UpdateArticleById">
        <PARAMETER name="ArticleId"/>
        <PARAMETER name="Title"/>
        <PARAMETER name="SubTitle"/>
        <PARAMETER name="Reporter"/>
        <PARAMETER name="Content"/>
    </method>
</public>

...

Const DB_CON = "File Name=d:\Config\www.goodnews.com.UDL"
Const adInteger = 3
Const adChar = 129
Const adLongVarChar = 201
Const adVarChar = 200

Function GetArticles()

...

Function UpdateArticleById( _
    strArticleId, strTitle, strSubTitle, strReporter, strContent)

```

End Function

```
' private functions -----  
...
```

:/admin/C.ArticleUpdateAction.asp

C.ArticleUpdateAction.asp

```
<%  
Option Explicit  
%>  
<!--#include virtual="/Common/Connection.asp" -->  
<!--#include virtual="/Common/Const.asp" -->  
<!--#include virtual="/Common/CDebug.asp" -->  
<!--#include virtual="/Common/Declare.asp" -->  
<!--#include virtual="/Common/Utils.asp" -->  
<%  
Debug.Enabled = False  
  
Call article.UpdateArticleById( _  
    Request("ArticleId"), _  
    Request("Title"), _  
    Request("SubTitle"), _  
    Request("Reporter"), _  
    Request("Content"))  
  
Call FreeObjects()  
Response.Redirect "ArticleList.asp"  
%>
```

GoodNews.Article.WSC

```
Function UpdateArticleById( _  
    strArticleId, strTitle, strSubTitle, strReporter, strContent)  
  
    Dim dbh, strSQL  
  
    strSQL = "UPDATE Articles SET " & _  
        " Title = ?, " & _  
        " SubTitle = ?, " & _  
        " Reporter = ?, " & _  
        " Content = ? " & _  
        " WHERE ArticleId = ?"  
  
    Set dbh = Server.CreateObject("DAL.DBHelper")  
    Call dbh.RunSQL( _  
        strSQL, _  
        Array( _  
            dbh.mp("@Title", adVarChar, 150, strTitle), _  
            dbh.mp("@SubTitle", adVarChar, 200, strSubTitle), _  
            dbh.mp("@Reporter", adVarChar, 20, strReporter), _  
            dbh.mp("@Content", adLongVarChar, Len(strContent), strContent), _  
            dbh.mp("@ArticleId", adInteger, 4, strArticleId) _
```

```
        ), DB_CON)  
        Set dbh = Nothing  
End Function
```